



an interpreted imperative programming
language that wants to believe

Lorenzo Loconte

Introduction

Shrimp is a very simple and didactic imperative programming language designed during the course of *Formal Methods for Computer Science* at Università degli Studi di Bari Aldo Moro. **Shrimp** uses an *eager evaluation strategy*. In order to ensure that, the interpreter executes the code using the *call by value* method.

Software Modules

The program is composed by three main components:

- The **parser**
- The **optimizer**
- The **interpreter**

The **parser** takes in input the source code and convert it into an intermediate representation. The intermediate representation have the structure of a n -ary tree having the non-terminals of the grammar as internal nodes and commands, identifiers and constants on the leaves.

The **optimizer** takes in input the intermediate representation given by the parser. The result of the optimizer is an *optimized* intermediate representation. It evaluates the constant expressions (both arithmetic and boolean) that might be present in the source code and replace them with the resulting constants. The optimizer also checks for empty command blocks and useless branch statements and removes (or optimize) them.

The **interpreter** execute the semantics present in an intermediate representation. The basic idea is to use a **state** (or environment) that collects the values of the variables during the execution of the program. The result of the interpretation is the resulting state, that is a set of ground assignments to the variables. The variables can be either of type integer, boolean or array of fixed size of integers.

Functors, Applicatives and Monads

Before introducing the syntax of the programming language, let's introduce three useful classes that are widely used in the construction of an interpreter. These classes are defined on “wrapped” types, such as `Maybe`.

The `Functor` class defines the `fmap` function. The `fmap` function takes in input a simple function `a -> b` and an object of type `a` wrapped inside the functor `f`. The result of the `fmap` function is the application of the function to the “unwrapped” object inside `f`.

```
1 class Functor f where
2   fmap :: (a -> b) -> f a -> f b
```

The `Applicative` class can be used only if `Functor` is already implemented for that polymorphic data type. It defines a function named `pure` and the `<*>` operator. The `pure` function simply takes in input an object and wraps it into an applicative `f`. The `<*>` operator is more complex. It takes in input a function `a -> b` wrapped in an applicative `f` and an object of type `a` wrapped inside the applicative `f`. The result of the `<*>` operator is the application of the “unwrapped” function to the “unwrapped” object inside `f`.

```
1 class (Functor f) => Applicative f where
2   pure :: a -> f a
3   (<*>) :: f (a -> b) -> f a -> f b
```

The last important class is the `Monad` class. The `Monad` class can be used only if `Applicative` is already implemented for that polymorphic data type. It defines a function named `return` and the `bind` operator. The `return` function is very similar to the `pure` function defined for `Applicatives`. The difference is that `return` gives an object wrapped inside a `Monad` while `pure` gives an object wrapped inside an `Applicative`. Often, their implementation is equivalent. The `bind` operator is denoted with `>=>` and takes in input an object of type `a` wrapped inside the monad `m` and a function `a -> m b`. The result of the `bind` operator is application of the function to the “unwrapped” object inside `m`.

```
1 class (Applicative m) => Monad m where
2   return :: a -> m a
3   return = pure
4   (>=>) :: m a -> (a -> m b) -> m b
```

The `Monad` class is very useful because it's possible to *emulate* the behavior of an imperative programming language inside Haskell, a purely functional programming language. That is, the following code snippet ...

```
1 m1 >=> \a1 ->
2   m2 >=> \a2 ->
3     ...
4     mn >=> \an ->
5       f a1 a2 ... an
```

... can be simplified using the `do` construct, as in a imperative programming language.

```
1 do
2   a1 <- m1
3   a2 <- m2
4   ...
5   an <- mn
6   f a1 a2 ... an
```

The Language Syntax

The syntax for the **Shrimp** programming language is a context-free grammar. So, it can be denoted using some kind of EBNF (Extended Backus Naur Form) as following:

```
1 Integer ::= [0-9]+
2 Identifier ::= [a-zA-Z_]+ \ Keyword
3 Program ::= Header "shrimp" ";" Block
4 Header ::= [Variable]*
5 Block ::= [Command]*
6
7 Variable ::= "let" Identifier "as" Type ";"
8 Type ::= {"int" | "bool" | "array" "[" Integer "]" }
9 Command ::= {Assignment | Branch | Loop | Skip} ";"
10
11 Assignment ::= Identifier
12   { "=" ArithmeticExpr
13   | "=" BooleanExpr
14   | "[" ArithmeticExpr "]" "=" ArithmeticExpr
15   }
16
17 Branch ::= "if" "(" BooleanExpr ")" "then" Block ["else" Block] "end"
18 Loop ::= "while" "(" BooleanExpr ")" "do" Block "end"
19 Skip ::= "skip"
20
21 ArithmeticExpr ::= ArithmeticTerm
22   [{"+" | "-"} ArithmeticTerm]*
23 ArithmeticTerm ::= ArithmeticFactor
24   [{"*" | "/" | "%"} ArithmeticFactor]*
25 ArithmeticFactor ::=
26   Integer
27   | Identifier
28   | Identifier "[" ArithmeticExpr "]"
29   | "-" ArithmeticExpr
30   | "(" ArithmeticExpr ")"
31
32 BooleanExpr ::= ArithmeticTerm ["or" ArithmeticTerm]*
33 BooleanTerm ::= ArithmeticFactor ["and" ArithmeticFactor]*
34 BooleanFactor ::=
35   "true"
36   | "false"
37   | Identifier
38   | "not" BooleanExpr
39   | ArithmeticExpr "eq" ArithmeticExpr
40   | ArithmeticExpr "neq" ArithmeticExpr
41   | ArithmeticExpr "lt" ArithmeticExpr
42   | ArithmeticExpr "gt" ArithmeticExpr
43   | ArithmeticExpr "leq" ArithmeticExpr
44   | ArithmeticExpr "geq" ArithmeticExpr
45   | "(" BooleanExpr ")"
```

Some of the non-terminals of this context-free grammar are reported directly in Haskell. That is, I defined an *abstract syntax tree* that also represents the intermediate representation of a program. This intermediate representation will be the result of the parser. Moreover, the presence of an intermediate representation permits us to apply optimizations *at prior* respect to the interpretation step. The following code snippet contains the definition of the *abstract syntax tree*.

```

1 type Program = (Header, Block)
2 type Block = [Command]
3 type Header = [Variable]
4
5 data ArithmeticExpr
6   = Constant Int
7   | IntegerVar String
8   | ArrayVar String ArithmeticExpr
9   | Add ArithmeticExpr ArithmeticExpr
10  | Sub ArithmeticExpr ArithmeticExpr
11  | Mul ArithmeticExpr ArithmeticExpr
12  | Div ArithmeticExpr ArithmeticExpr
13  | Mod ArithmeticExpr ArithmeticExpr
14  | Neg ArithmeticExpr
15  deriving (Eq, Show)
16
17 data BooleanExpr
18   = Truth Bool
19   | BooleanVar String
20   | Not BooleanExpr
21   | Or BooleanExpr BooleanExpr
22   | And BooleanExpr BooleanExpr
23   | Equal ArithmeticExpr ArithmeticExpr
24   | NotEqual ArithmeticExpr ArithmeticExpr
25   | Less ArithmeticExpr ArithmeticExpr
26   | Greater ArithmeticExpr ArithmeticExpr
27   | LessEqual ArithmeticExpr ArithmeticExpr
28   | GreaterEqual ArithmeticExpr ArithmeticExpr
29   deriving (Eq, Show)
30
31 data Command
32   = Skip
33   | ArithmeticAssignment String ArithmeticExpr
34   | BooleanAssignment String BooleanExpr
35   | ArrayAssignment String ArithmeticExpr ArithmeticExpr
36   | Branch BooleanExpr Block Block
37   | Loop BooleanExpr Block
38   deriving (Eq, Show)
39
40 data Variable
41   = IntegerDecl String
42   | BooleanDecl String
43   | ArrayDecl String Int
44   deriving (Eq, Show)

```

The Parser

The parser can be viewed as a function from a string to a list of pairs of values and strings (Graham Hutton).

```

1 newtype Parser a = Parser {unwrap :: String -> [(a, String)]}

```

Note that the parser have a special function called `unwrap` that takes the function out from the parser. The parser is implemented in a “monadic” way. That is, I implemented the following interfaces: *monad plus* and *monad alternative* as suggested in *Monadic*

Parsing in Haskell (Hutton & Meijer). In order to implement the monad interface I also implemented the *functor*, *applicative* and *monad* interfaces.

```
1 instance Functor Parser where
2   fmap f p = Parser (\cs ->
3     [(f a, cs') | (a, cs') <- unwrap p cs])
```

The functor implementation for the parser implements the `fmap` function, that is the application of a function on a wrapped parser.

```
1 instance Applicative Parser where
2   pure a = Parser (\cs -> [(a, cs)])
3   p <*> q = Parser (\cs -> concat
4     [unwrap (fmap a q) cs' | (a, cs') <- unwrap p cs])
```

Moreover, the applicative implementation for the parser introduces both the function `pure` and the operator `<*>`. The `pure` function takes a simple value and wraps it into a parser. The `<*>` operator takes in input a function wrapped in a parser and another parser. The result is the application of the wrapped function onto the parser.

```
1 instance Monad Parser where
2   return a = pure a
3   p >=> f = Parser (\cs -> concat
4     [unwrap (f a) cs' | (a, cs') <- unwrap p cs])
```

The `bind` operator `>=>` takes as input a parser and a function that returns a parser. The result of this operator is the application of the function to the “unwrapped” parser. The application of multiple `bind` nested operations can be abbreviated using the `do` construct in Haskell.

Moreover, two other *custom* interfaces are implemented, needed for making the parser: `MonadPlus` and `MonadAlternative`. The `plus` operator defined on parsers concatenates the result of each one. Moreover, the `MonadAlternative` give us an operator `<|>` useful for combining parsers in a mutually exclusive way. The classes definition can be found in the following code snippet.

```
1 class (Monad m) => MonadPlus m where
2   zero :: m a
3   plus :: m a -> m a -> m a
4
5 class (MonadPlus m) => MonadAlternative m where
6   (<|>) :: m a -> m a -> m a
7   many :: m a -> m [a]
8   many m = some m <|> return []
9   some :: m a -> m [a]
10  some m = liftA2 (:) m (many m)
11  chain :: m a -> m (a -> a -> a) -> m a
12  chain p o = do a <- p; rest a
13    where
14      rest a = (do f <- o; a' <- p; rest (f a a')) <|> return a
```

Note that the function `liftA2` is the composition of `fmap` and the applicative operator `<*>`. The functions `many` and `some` are called *combinators* and are used to define the concept of repeated parsing. The `many` function refers to zero to any number of applications of a parser while the `some` function refers to at least one to any number of applications of a parser. The implementation for the parser of the `bind` operator and the other functions described before is the following. The `chain` function is another combinator that it’s used for left-associative recursion on parsers, as described by Hutton & Meijer.

```

1 instance MonadPlus Parser where
2   zero = Parser (const [])
3   p `plus` q = Parser (\cs -> unwrap p cs ++ unwrap q cs)
4
5 instance MonadAlternative Parser where
6 (<|>) p q = Parser (\cs ->
7   case unwrap (p `plus` q) cs of
8     [] -> []
9     (x : _) -> [x])

```

The `zero` function defines what is an empty parser. Note that in this context an empty parser is a *failed* parser, that is a parser that results from a syntactical error. The `plus` function concatenates the results of two individual parsers. Using the monadic definition of parser, it permits us to easily build a parser for each structure of the programming language, without having to handle the eventual parsing errors individually. For example, it's possible to define a parser for symbols, identifiers and integers as in the following code snippet. Moreover, since the parser implements the `MonadAlternative` interface described above, the combination of multiple parsers is straightforward and easily *parallelizable* by the compiler.

```

1 item :: Parser Char
2 item = Parser (\case "" -> []; (c : cs) -> [(c, cs)])
3
4 satisfy :: (Char -> Bool) -> Parser Char
5 satisfy p = do c <- item; if p c then return c else zero
6
7 token :: Parser a -> Parser a
8 token p = do space; v <- p; space; return v
9
10 space :: Parser String
11 space = many $ satisfy isSpace
12
13 identifier :: Parser String
14 identifier = do
15   s <- token $ some $ satisfy isLetter
16   if s `elem` keywords
17     then zero
18     else return s
19
20 constant :: Parser Int
21 constant = read <$> token (some $ satisfy isDigit)
22
23 char :: Char -> Parser Char
24 char c = satisfy (c ==)

```

That is, I firstly defined a `item` function that reads a character from the string. Then I defined a function called `satisfy` that apply a constraint to the character read by the parser. This function returns an `zero` parser if the constraint is not satisfied. In the end, using both the combinators `many` and `some`, I built parsers for identifiers, constants and also spaces that are the main blocks of the context-free grammar already defined. In the same way, it's possible to define a parser for other constructs, such as keywords and symbols. It's important to note that the identifier parser fails if a keyword is found. This is a very fundamental aspect for a programming language. So, it's not possible use variables identified by keywords.

```

1 keyword :: String -> Parser String
2 keyword cs = token $ word cs
3
4 word :: String -> Parser String
5 word [c] = do char c; return [c]
6 word (c : cs) = do char c; word cs; return (c : cs)
7
8 symbol :: Char -> Parser Char
9 symbol c = token $ char c

```

In order to build parsers for both arithmetic and boolean expression the `<|>` operator between parsers is used. So, I defined the parser for arithmetic expressions using other sub-parsers that are used in order to maintain the operators precedence. Moreover, I used the `chain` combinator in order to build the *abstract syntax tree* by the left-associative operators of the grammar.

```

1 arithmeticExpr :: Parser ArithmeticExpr
2 arithmeticExpr = chain arithmeticTerm op
3   where
4     op =
5       do symbol '+'; return Add
6       <|> do symbol '-'; return Sub
7
8 arithmeticTerm :: Parser ArithmeticExpr
9 arithmeticTerm = chain arithmeticFactor op
10  where
11    op =
12      do symbol '*'; return Mul
13      <|> do symbol '/'; return Div
14      <|> do symbol '%'; return Mod
15
16 arithmeticFactor :: Parser ArithmeticExpr
17 arithmeticFactor =
18   do Constant <$> constant
19   <|> do
20     d <- identifier
21     do
22       symbol '['
23       k <- arithmeticExpr
24       symbol ']'
25       return (ArrayVar d k)
26     <|> return (IntegerVar d)
27   <|> do symbol '-'; Neg <$> arithmeticExpr
28   <|> do symbol '('; a <- arithmeticExpr; symbol ')'; return a

```

The `chain` combinator is used in both `arithmeticExpr` and `arithmeticTerm` in order to permits to concatenate multiple addition/subtraction operators and multiple multiplication/division/modulus operators. If we didn't used the `chain` combinator but simple plain tail recursion we would have that the order of operations would be from right to left instead of left to right. The parser for boolean expressions is similar to the parser for arithmetic expressions and so it's omitted in this documentation.

In the end we have a parser for every command described in the grammar. It's important to notice that the grammar also allows for `if-then` statements (i.e. without the `else` command block). A parser that works on both `if-then-else` and `if-then` statements is implemented by combining the two individual parsers using the `<|>` operator.

Moreover, using the many combinator and the `<|>` operator, the parsers for both multiple commands and a single command are straightforward.

```

1 command :: Parser Command
2 command = do
3   c <- assignment <|> branch <|> loop <|> skip
4   symbol ';'
5   return c
6
7 assignment :: Parser Command
8 assignment = do
9   d <- identifier
10  do
11    symbol '='
12    ArithmeticAssignment d <$> arithmeticExpr
13    <|> do
14      symbol '='
15      BooleanAssignment d <$> booleanExpr
16    <|> do
17      symbol '['
18      k <- arithmeticExpr
19      symbol ']'
20      symbol '='
21      ArrayAssignment d k <$> arithmeticExpr
22
23 branch :: Parser Command
24 branch = do
25   keyword "if"
26   symbol '('
27   b <- booleanExpr
28   symbol ')'
29   keyword "then"
30   c1 <- block
31   do
32     keyword "else"
33     c2 <- block
34     keyword "end"
35     return (Branch b c1 c2)
36   <|> do
37     keyword "end"
38     return (Branch b c1 [Skip])
39
40 loop :: Parser Command
41 loop = do
42   keyword "while"
43   symbol '('
44   b <- booleanExpr
45   symbol ')'
46   keyword "do"
47   c <- block
48   keyword "end"
49   return (Loop b c)
50
51 skip :: Parser Command
52 skip = do
53   keyword "skip"
54   return Skip

```


Another parser that is needed for the grammar, is the parser for the variable declarations. That is, in this programming language we can declare variables only on the top of the program, and before the `shrimp` keyword. The variables can be integers, boolean or array of integers. For the declaration of array of integers, the size of the array must be fixed and a constant integer value.

```

1 variable :: Parser Variable
2 variable = do
3   keyword "let"
4   d <- identifier
5   keyword "as"
6   v <- var
7   symbol ';'
8   return (v d)
9   where
10    var =
11      do keyword "int"; return IntegerDecl
12      <|> do keyword "bool"; return BooleanDecl
13      <|> do
14        keyword "array"
15        symbol '['
16        n <- constant
17        symbol ']'
18        return (`ArrayDecl` n)

```

The parser for the entire program is defined as in the following code snippet. That is, the parser for a block of commands is defined using the `many` combinator on the parser for a command. In the same way, the parser for a header is defined using the `many` combinator on the parser for a variable. The parser for the entire parser is defined using both the parser for the header and the command block. Note that the keyword `shrimp` separates these parts of the program.

```

1 program :: Parser Program
2 program = do
3   h <- header
4   keyword "shrimp"
5   symbol ';'
6   b <- block
7   return (h, b)
8
9 block :: Parser Block
10 block = many command
11
12 header :: Parser Header
13 header = many variable
14
15 parse :: String -> Result (Program, String)
16 parse cs = case unwrap program cs of
17   [] -> Error EmptyProgram
18   [(p, cs)] -> Ok (p, cs)

```

The Optimizer

The *optimization* process is an intermediate step between the parsing and the interpretation of the program itself. Currently, the main optimization step is related to the execution of constant values. That is, if an expression in a loop is defined only on constant values, it's better to optimize the computation of that expression by replacing it with its constant result. This procedure is done before the interpretation of the program. The implementation of the *optimization* step in Haskell is straightforward, due to simple recursion functions.

For example, consider the following arithmetic expression, expressed in intermediate representation, that we wish to optimize. If we apply the optimization step to this arithmetic expression, we obtain the equivalent but more efficient arithmetic expression. A very similar optimization process is also implemented on boolean expressions.

```
1 let expr = Div (  
2   (Mul (Identifier "x") (Sub (Constant 5) (Constant 3)))  
3   (Add (Constant 9) (Constant 1)))
```

```
1 let expr' = Div (Mul (Identifier "x") (Constant 2)) (Constant 10)
```

The *optimization* process also includes a basic optimization on commands such as **skip**, **if-then-else** and **while-do** statements. First of all, all the **skip** commands are removed from the intermediate representation. Moreover, if the condition of a **if-then-else** command is always *true* then the entire statement is replaced with the first block of commands. In a similar way, if the condition is always *false* then the entire statement is replaced with the second block of commands.

```
1 let command = Branch (Equal (Sub (Constant 1) (Constant 1)) 0)  
2   [Assignment "x" (Constant 1)]  
3   [Assignment "x" (Constant 2)]
```

```
1 let command' = [Assignment "x" (Constant 1)]
```

Furthermore, this approach is also used for **while-do** commands. That is, if the condition of a **while-do** command is always *false* then the entire statement is completely removed.

```
1 let command = Loop (Equal (Sub (Constant 1) (Constant 1)) 1)  
2   [Assignment "x" (Add (IntegerVar "x") (Constant 1))]
```

```
1 let command' = Skip
```

However, if the condition is always *true*, an exception named *Infinite Loop* is raised. In other words, the optimizer is capable of detecting trivial infinite loops and prevents the interpretation of such programs. For example, the following loop program's intermediate representation is not interpreted, because the *Infinite Loop* exception is raised *before* the interpretation step.

```
1 let command = Loop (Equal (Sub (Constant 1) (Constant 1)) 0)  
2   [Assignment "x" (Add (IntegerVar "x") (Constant 1))]
```

These kinds of code optimizations are very basic and don't consider the *expected state* of the program at a certain point, in order to apply more advanced kinds of optimizations (such as expressions simplification or variables pruning).

The Interpreter

Finally, the obtained intermediate representation is interpreted by the interpreter. The interpreter scan the intermediate representation using *depth-first traversal* on the n -ary tree that compose the intermediate representation itself. Before introducing the execution of the interpreter on an intermediate representation, some useful data types are defined in order to handle errors.

```
1 data Exception
2   = EmptyProgram
3   | InfiniteLoop
4   | DivisionByZero
5   | UndeclaredVariable String
6   | MultipleVariable String
7   | TypeMismatch String
8   | OutOfBound String Int
9   | InvalidSize String
10
11 instance Show Exception where
12   show EmptyProgram = "Empty Program"
13   show InfiniteLoop = "Infinite Loop"
14   show DivisionByZero = "Division By Zero"
15   show (UndeclaredVariable d) = "Undeclared Variable" ++ ": " ++ d
16   show (MultipleVariable d) = "Multiple Variable" ++ ": " ++ d
17   show (TypeMismatch d) = "Type Mismatch" ++ ": " ++ d
18   show (OutOfBound d i) = "Out Of Bound" ++ ": " ++ d ++ " at " ++ show i
19   show (InvalidSize d) = "Invalid Size" ++ ": " ++ d
20
21 data Result a = Ok a | Error Exception
22
23 instance Functor Result where
24   fmap f (Ok v) = Ok (f v)
25   fmap _ (Error e) = Error e
26
27 instance Applicative Result where
28   pure v = Ok v
29   (<*>) (Ok f) (Ok v) = Ok (f v)
30   (<*>) (Error e) _ = Error e
31   (<*>) _ (Error e) = Error e
32
33 instance Monad Result where
34   (>>=) (Ok v) f = f v
35   (>>=) (Error e) _ = Error e
36
37 exception :: Exception -> a
38 exception e = errorWithoutStackTrace $ show e
```

As one can see, the `Result` data type is a polymorphic type that can be either a `Ok` or a `Error`. The `Error` type also encapsulate an exception, one of the listed above. Moreover functor, applicative and monad interfaces are implemented in order to apply the needed operators directly on intermediate results. The functor, applicative and monad interfaces implementation is as in the `Maybe` data type. In the end, the `exception` function is useful in order to print the error and stop the program without printing the stack trace.

In order to interpret the intermediate representation, two sub-interpreters are implemented. The first interpreter loads all the variables in the variables declaration section of the program. The result of the variables loading is a consistent state, viewed as a dictio-

nary with key as the identifier of a variable and value the effective value of the variable at a certain point of the program. The value of a variable can be either an integer, a boolean value or a vector of integers.

```

1 initialize :: State -> Header -> State
2 initialize s [] = s
3 initialize s ((IntegerDecl d) : hs) =
4     case search d s of
5         Just _ -> exception (MultipleVariable d)
6         Nothing -> initialize s' hs
7         where
8             s' = insert d (IntegerValue 0) s
9 initialize s ((BooleanDecl d) : hs) =
10     case search d s of
11         Just _ -> exception (MultipleVariable d)
12         Nothing -> initialize s' hs
13         where
14             s' = insert d (BooleanValue False) s
15 initialize s ((ArrayDecl d n) : hs) =
16     case search d s of
17         Just _ -> exception (MultipleVariable d)
18         Nothing ->
19             if n > 0
20             then initialize s' hs
21             else exception (InvalidSize d)
22         where
23             s' = insert d (ArrayValue (zeroArray n)) s

```

The second interpreter executes the commands specified in the commands section of the command (after the `shrimp` keyword). So, it interprets every command specified in the grammar, that are `skip`, various kinds of `assignments`, `if-then-else` and `if-then` statements and the `while-do` statement.

```

1 execute :: State -> Block -> State
2 execute s [] = s
3 execute s (Skip : cs) = execute s cs
4 execute s ((ArithmeticAssignment d a) : cs) =
5     case search d s of
6         Just (IntegerValue _) ->
7             case evalArithmetic s a of
8                 Ok v -> execute s' cs
9                 where
10                     s' = insert d (IntegerValue v) s
11             Error e -> exception e
12         Just (BooleanValue _) -> exception (TypeMismatch d)
13         Just (ArrayValue _) -> exception (TypeMismatch d)
14         Nothing -> exception (UndeclaredVariable d)
15 execute s ((BooleanAssignment d b) : cs) =
16     case search d s of
17         Just (BooleanValue _) ->
18             case evalBoolean s b of
19                 Ok t -> execute s' cs
20                 where
21                     s' = insert d (BooleanValue t) s
22             Error e -> exception e
23         Just (IntegerValue _) -> exception (TypeMismatch d)
24         Just (ArrayValue _) -> exception (TypeMismatch d)
25         Nothing -> exception (UndeclaredVariable d)

```

```

26 execute s ((ArrayAssignment d k a) : cs) =
27   case search d s of
28     Just (ArrayValue vs) ->
29       case (evalArithmetic s a, evalArithmetic s k) of
30         (Ok v, Ok i) -> execute s' cs
31         where
32           s' = insert d (ArrayValue vs') s
33           vs' = case writeArray i v vs of
34             Just vs' -> vs'
35             Nothing -> exception (OutOfBounds d i)
36         (Error e, _) -> exception e
37         (_, Error e) -> exception e
38     Just (IntegerValue _) -> exception (TypeMismatch d)
39     Just (BooleanValue _) -> exception (TypeMismatch d)
40     Nothing -> exception (UndeclaredVariable d)
41 execute s ((Branch b cs' cs'') : cs) =
42   case evalBoolean s b of
43     Ok True -> execute s (cs' ++ cs)
44     Ok False -> execute s (cs'' ++ cs)
45     Error e -> exception e
46 execute s (c@(Loop b cs') : cs) =
47 execute s (Branch b (cs' ++ [c]) [Skip] : cs)

```

The implementation of the evaluation of arithmetic expressions make *heavily* use of the methods exposed by the applicative interface. Moreover, the `liftA2` function is used in order to make the evaluation functions implementation more compact and readable. It's important to notice that arithmetic expressions evaluation includes reading from variables and accessing to arrays by an index that is the evaluation of another arithmetic expression. Moreover, the current implementation includes the evaluation of the `minus` unary operator. In this implementation, I also used the functions `safeDiv`, `safeMod` and `seqM2`. In other words, the `seqM2` is a custom function that implements binary sequencing, similarly to the `bind` operator.

```

1 evalArithmetic :: State -> ArithmeticExpr -> Result Int
2 evalArithmetic _ (Constant v) = Ok v
3 evalArithmetic s (IntegerVar d) =
4   case search d s of
5     Just (IntegerValue v) -> Ok v
6     Just (BooleanValue _) -> exception (TypeMismatch d)
7     Just (ArrayValue _) -> exception (TypeMismatch d)
8     Nothing -> exception (UndeclaredVariable d)
9 evalArithmetic s (ArrayVar d k) =
10  case search d s of
11    Just (ArrayValue vs) ->
12      case evalArithmetic s k of
13        Ok i ->
14          case readArray i vs of
15            Just v -> Ok v
16            Nothing -> Error (OutOfBounds d i)
17        Error e -> exception e
18    Just (IntegerValue _) -> exception (TypeMismatch d)
19    Just (BooleanValue _) -> exception (TypeMismatch d)
20    Nothing -> exception (UndeclaredVariable d)
21 evalArithmetic s (Add a1 a2) = liftA2 (+) v1 v2
22 where
23   v1 = evalArithmetic s a1
24   v2 = evalArithmetic s a2

```

```

25 evalArithmetic s (Sub a1 a2) = liftA2 (-) v1 v2
26   where
27     v1 = evalArithmetic s a1
28     v2 = evalArithmetic s a2
29 evalArithmetic s (Mul a1 a2) = liftA2 (*) v1 v2
30   where
31     v1 = evalArithmetic s a1
32     v2 = evalArithmetic s a2
33 evalArithmetic s (Div a1 a2) = seqM2 safeDiv v1 v2
34   where
35     v1 = evalArithmetic s a1
36     v2 = evalArithmetic s a2
37 evalArithmetic s (Mod a1 a2) = seqM2 safeMod v1 v2
38   where
39     v1 = evalArithmetic s a1
40     v2 = evalArithmetic s a2
41 evalArithmetic s (Neg a) = negate <$> v
42   where
43     v = evalArithmetic s a

```

The implementations of `safeDiv`, `safeMod`, and `seqM2` functions are listed in the following code snippet. Note that `safeDiv` and `safeMod` are special functions that handles divisions by zero at runtime.

```

1 safeDiv :: Int -> Int -> Result Int
2 safeDiv _ 0 = Error DivisionByZero
3 safeDiv u v = Ok (div u v)
4
5 safeMod :: Int -> Int -> Result Int
6 safeMod _ 0 = Error DivisionByZero
7 safeMod u v = Ok (mod u v)
8
9 seqM2 :: (Monad m) => (a -> b -> m c) -> m a -> m b -> m c
10 seqM2 f x y = join $ liftA2 f x y
11
12 join :: (Monad m) => m (m a) -> m a
13 join m = m >>= id

```

The implementation of the evaluation of boolean expressions is very similar to the one for arithmetic expressions, so it's omitted.

Conclusion

The monadic implementation of the parser give us a very simple way of concatenating and combining multiple parsers of sub-grammars. Also, it's very easy to extend the grammar of the language in order to include other commands or statements. Moreover, the use of an intermediate representation permits us to apply post-processing and optimizations and make the interpretation itself straightforward and more efficient.

Future works may include the introduction of other types for variables. Another extension of this work consists of adding useful information about the parsing errors, i.e. missing tokens and relevant row and column locations in the source code where the error occurred. However, other improvements can be done in the optimization step. That is, one can implement more “aggressive” optimizations of the intermediate representation based on the expected state of memory of the program during its execution.

Running an example

First of all, navigate to the project directory. Then, open `ghci` and load the needed modules as following.

```
1 :load Main.hs Shrimp.hs
```

Finally, run the main program and insert the path of a source file in the `examples` directory (for example `primes.shr`).

```
1 *Main> main
2 The Shrimp Interpreter
3 Insert the path of the source file:
4 examples/primes.shr
5 Memory state:
6     "n": int = 49
7     "i": int = 15
8     "j": int = 0
9     "stop": bool = False
10    "primes": array[15] = [2,3,5,7,11,13,17,19,23,29,31,37,41,43,47]
```