



an interpreted imperative programming  
language that wants to believe

Lorenzo Loconte

## Introduction

**Shrimp** is a simple imperative programming language designed during the course of *Formal Methods for Computer Science* at Università degli Studi di Bari Aldo Moro. **Shrimp** uses an *eager evaluation strategy*. In order to ensure that, the interpreter executes the code using the *call by value* method.

## Software Modules

The program is composed by three main components:

- The **parser**
- The **optimizer**
- The **interpreter**

The **parser** takes in input the source code and convert it to an intermediate representation. The intermediate representation (IR) have the structure of a n-ary tree having the non-terminals of the grammar as internal nodes and commands, identifiers and constants on the leaves.

The **optimizer** takes in input the intermediate representation given by the parser. The result of the optimizer is an *optimized* intermediate representation. It evaluates the constant expressions (both arithmetic and boolean) that might be present in the source code and replace them with the resulting constants. The optimizer also checks for empty commands block and useless branch statements.

The **interpreter** execute the semantics present in an intermediate representation. The basic idea is to use a **state** (or environment) that collects the values of the variables during the execution of the program. The result of the interpretation is the resulting state, that is a set of assignments to the variables.

# The Language Syntax

The syntax for the **Shrimp** programming language is a context-free grammar. So, it can be denoted using EBNF (Extended Backus Naur Form) as following:

```
1 Type ::= "int"
2 Integer ::= [0-9]+
3 Identifier ::= [a-zA-Z_]+
4 Program ::= "shrimp" Block
5 Block ::= [Command]*
6 Command ::= {Assignment | Branch | Loop}
7 Assignment ::= Identifier "=" ArithmeticExpr ";"
8 Branch ::= "if" "(" BooleanExpr ")" "then" Block
9           ["else" Block] "end if" ";"
10 Loop ::= "while" "(" BooleanExpr ")" "do"
11         Block "end while" ";"
12
13 ArithmeticExpr ::=
14     ArithmeticTerm "+" ArithmeticExpr
15     | ArithmeticTerm "-" ArithmeticExpr
16     | ArithmeticTerm
17 ArithmeticTerm ::=
18     ArithmeticFactor "*" ArithmeticTerm
19     | ArithmeticFactor "/" ArithmeticTerm
20     | ArithmeticFactor "%" ArithmeticTerm
21     | ArithmeticFactor
22 ArithmeticFactor ::=
23     Integer
24     | Identifier
25     | "-" ArithmeticExpr
26     | "(" ArithmeticExpr ")"
27
28 BooleanExpr ::=
29     BooleanTerm "or" BooleanExpr
30     | BooleanTerm
31 BooleanTerm ::=
32     BooleanFactor "and" BooleanTerm
33     | BooleanFactor
34
35 BooleanFactor ::=
36     "true"
37     | "false"
38     | "not" BooleanExpr
39     | ArithmeticExpr "eq" ArithmeticExpr
40     | ArithmeticExpr "neq" ArithmeticExpr
41     | ArithmeticExpr "lt" ArithmeticExpr
42     | ArithmeticExpr "gt" ArithmeticExpr
43     | ArithmeticExpr "leq" ArithmeticExpr
44     | ArithmeticExpr "geq" ArithmeticExpr
45     | "(" BooleanExpr ")"
```

Some of the non-terminals of this context-free grammar are reported directly in Haskell. That is, I defined an *abstract syntax tree* that also represents the intermediate representation of a program. This intermediate representation will be the result of the parser. Moreover, the presence of an intermediate representation permits us to apply optimizations *at prior* respect to the interpretation step. The following code snippet contains the definition of the *abstract syntax tree*.

```

1 data ArithmeticExpr
2   = Add ArithmeticExpr ArithmeticExpr
3   | Sub ArithmeticExpr ArithmeticExpr
4   | Mul ArithmeticExpr ArithmeticExpr
5   | Div ArithmeticExpr ArithmeticExpr
6   | Mod ArithmeticExpr ArithmeticExpr
7   | Neg ArithmeticExpr
8   | Constant Int
9   | Identifier String
10  deriving (Eq, Show)
11
12 data BooleanExpr
13   = Boolean Bool
14   | Not BooleanExpr
15   | Or BooleanExpr BooleanExpr
16   | And BooleanExpr BooleanExpr
17   | Equal ArithmeticExpr ArithmeticExpr
18   | NotEqual ArithmeticExpr ArithmeticExpr
19   | Less ArithmeticExpr ArithmeticExpr
20   | Greater ArithmeticExpr ArithmeticExpr
21   | LessEqual ArithmeticExpr ArithmeticExpr
22   | GreaterEqual ArithmeticExpr ArithmeticExpr
23  deriving (Eq, Show)
24
25 data Command
26   = Skip
27   | Assignment String ArithmeticExpr
28   | Branch BooleanExpr Block Block
29   | Loop BooleanExpr Block
30  deriving (Eq, Show)
31
32 type Block = [Command]

```

## The Parser

The parser can be viewed as a function from a string to a list of pairs of values and strings (Graham Hutton).

```

1 newtype Parser a = Parser {unwrap :: String -> [(a, String)]}

```

Note that the parser have a special function called `unwrap` that takes the function out from the parser. The parser is implemented in a “monadic” way. That is, I implemented the following interfaces: *monad plus* and *monad alternative* as suggested in *Monadic Parsing in Haskell* (Hutton & Meijer). In order to implement the monad interface I also implemented the *functor* and *applicative* interfaces as in the following code snippet.

```

1 instance Functor Parser where
2   fmap f p = Parser (\cs ->
3     [(f a, cs') | (a, cs') <- unwrap p cs])
4
5 instance Applicative Parser where
6   pure a = Parser (\cs -> [(a, cs)])
7   p <*> q = Parser (\cs -> concat
8     [unwrap (fmap a q) cs' | (a, cs') <- unwrap p cs])

```

The functor implementation for the parser implements the `fmap` function, that is the application of a function on a wrapped parser. Moreover, the applicative implementation for the parser implement both the function `pure` and the operator `<*>`. The `pure` function takes a simple values and wraps it into a parser. The `<*>` operator takes in input a function wrapped in a parser and another parser. The result is the application of the wrapped function onto the parser.

The next step is to implement the standard monad interface. Moreover, two more interfaces are implemented: `MonadPlus` and `MonadAlternative`. The `plus` operator defined on parsers concatenates the result of each one. Moreover, the `MonadAlternative` give us an operator `<|>` useful for combining parsers in a mutually exclusive way. The classes definition can be found in the following code snippet.

```

1 class (Monad m) => MonadPlus m where
2 zero :: m a
3 plus :: m a -> m a -> m a
4
5 class (MonadPlus m) => MonadAlternative m where
6 (<|>) :: m a -> m a -> m a
7 many :: m a -> m [a]
8 many m = some m <|> return []
9 some :: m a -> m [a]
10 some m = liftA2 (:) m (many m)

```

Note that the function `liftA2` is the composition of `fmap` and the applicative operator `<*>`. The functions `many` and `some` are called *combinators* and are used to define the concept of repeated parsing. The `many` function refers to zero to any number of applications of a parser while the `some` function refers to at least one to any number of applications of a parser. The implementation for the parser of the monad sequencing and the other functions and operators described before is the following.

```

1 instance Monad Parser where
2 return a = pure a
3 p >= f = Parser (\cs -> concat
4   [unwrap (f a) cs' | (a, cs') <- unwrap p cs])
5
6 instance MonadPlus Parser where
7 zero = Parser (const [])
8 p `plus` q = Parser (\cs -> unwrap p cs ++ unwrap q cs)
9
10 instance MonadAlternative Parser where
11 (<|>) p q = Parser (\cs ->
12   case unwrap (p `plus` q) cs of
13     [] -> []
14     (x : _) -> [x])

```

The `zero` operator defines what is an empty parser. Note that in this context an empty parser is a *failed* parser, that is a parser that results from a syntactical error. The `plus` operator concatenates the results of two individual parsers. Using the monadic definition of parser, it permits us to easily build a parser for each structure of the programming language, without having to handle the eventual parsing errors individually. For example, it's possible to define a parser for symbols, identifiers and integers as in the following code snippet. Moreover, since the parser implements the `MonadAlternative` interface described above, the combination of multiple parsers is straightforward and easily parallelizable.

```

1 item :: Parser Char
2 item = Parser (\case " " -> []; (c : cs) -> [(c, cs)])
3
4 satisfy :: (Char -> Bool) -> Parser Char
5 satisfy p = do c <- item; if p c then return c else zero
6
7 token :: Parser a -> Parser a
8 token p = do space; v <- p; space; return v
9
10 space :: Parser String
11 space = many $ satisfy isSpace
12
13 identifier :: Parser String
14 identifier = token $ some $ satisfy isLetter
15
16 constant :: Parser Int
17 constant = read <$> token (some $ satisfy isDigit)
18
19 char :: Char -> Parser Char
20 char c = satisfy (c ==)

```

That is, I firstly defined a `item` function that reads a character from the string. Then I defined a function called `satisfy` that apply a constraint to the character read by the parser. This function returns an `zero` parser if the constraint is not satisfied. In the end, using both the combinators `many` and `some`, I built parsers for identifiers, constants and also spaces that are the main blocks of the context-free grammar already defined. In the same way, it's possible to define a parser for other constructs, such as keywords and symbols.

```

1 keyword :: String -> Parser String
2 keyword cs = token $ word cs
3
4 word :: String -> Parser String
5 word [c] = do char c; return [c]
6 word (c : cs) = do char c; word cs; return (c : cs)
7
8 symbol :: Char -> Parser Char
9 symbol c = token $ char c

```

In order to build parsers for both arithmetic and boolean expression the `<|>` operator between parsers is used. So, I define the parser for arithmetic expressions using other sub-parsers that are used in order to maintain the operators precedence.

```

1 arithmeticExpr :: Parser ArithmeticExpr
2 arithmeticExpr = do
3   a <- arithmeticTerm
4   do symbol '+'; Add a <$> arithmeticExpr
5   <|> do symbol '-'; Sub a <$> arithmeticExpr
6   <|> do return a
7
8 arithmeticTerm :: Parser ArithmeticExpr
9 arithmeticTerm = do
10  a <- arithmeticFactor
11  do symbol '*'; Mul a <$> arithmeticTerm
12  <|> do symbol '/'; Div a <$> arithmeticTerm
13  <|> do symbol '%'; Mod a <$> arithmeticTerm
14  <|> do return a

```

```

15
16 arithmeticFactor :: Parser ArithmeticExpr
17 arithmeticFactor =
18   do Constant <$> constant
19   <|> do Identifier <$> identifier
20   <|> do symbol '-'; Neg <$> arithmeticExpr
21   <|> do symbol '('; a <- arithmeticExpr; symbol ')'; return a

```

The parser for boolean expressions is similar to the parser for arithmetic expressions and so it's omitted in this documentation. In the end we have a parser for every command described in the grammar. So, I defined a parser for the following commands: `skip`, `assignment`, `if-then-else`, and `while-do`.

```

1 assignment :: Parser Command
2 assignment = do
3   d <- identifier
4   symbol '='
5   a <- arithmeticExpr
6   symbol ';'
7   return (Assignment d a)
8
9 branch :: Parser Command
10 branch = do
11   keyword "if"
12   symbol '('
13   b <- booleanExpr
14   symbol ')'
15   keyword "then"
16   c1 <- block
17   do
18     keyword "else"
19     c2 <- block
20     keyword "end if"
21     symbol ';'
22     return (Branch b c1 c2)
23   <|> do
24     keyword "end if"
25     symbol ';'
26     return (Branch b c1 [Skip])
27
28 loop :: Parser Command
29 loop = do
30   keyword "while"
31   symbol '('
32   b <- booleanExpr
33   symbol ')'
34   keyword "do"
35   c <- block
36   keyword "end while"
37   symbol ';'
38   return (Loop b c)
39
40 skip :: Parser Command
41 skip = do
42   keyword "skip"
43   symbol ';'
44   return Skip

```

It's important to notice that the grammar also allows for `if-then` statements (i.e. without the `else` command block). A parser that works on both `if-then-else` and `if-then` statements is implemented by combining the two individual parsers using the `<|>` operator. Moreover, using the `many` combinator and the `<|>` operator, the parsers for both multiple commands and a single command are straightforward.

```
1 block :: Parser Block
2 block = many command
3
4 command :: Parser Command
5 command = assignment <|> branch <|> loop <|> skip
```

The parser for the entire program is defined as in the following code snippet.

```
1 program :: Parser Block
2 program = do keyword "shrimp"; block
3
4 parse :: String -> Result (Block, String)
5 parse cs = case unwrap program cs of
6   [] -> Error EmptyProgram
7   [(b, cs)] -> Ok (b, cs)
```

## The Optimizer

The *optimization* process is an intermediate step between the parsing and the interpretation of the program itself. Currently, the main optimization step is related to the execution of constant values. That is, if an expression in a loop is defined only on constant values, it's better to optimize the computation of that expression by replacing it with the result. This procedure is done before the interpretation of the program. For example, consider the following arithmetic expression, expressed in intermediate representation, that we wish to optimize.

```
1 let expr = Div (
2   (Mul (Identifier "x") (Sub (Constant 5) (Constant 3)))
3   (Add (Constant 9) (Constant 1)))
```

If we apply the optimization step to this arithmetic expression, we obtain the equivalent but more efficient arithmetic expression.

```
1 let expr' = Div (Mul (Identifier "x") (Constant 2)) (Constant 10)
```

The implementation of the *optimization* process in Haskell is straightforward, due to simple recursion functions. A very similar optimization process is also implemented on boolean expressions. The *optimization* process also includes a basic optimization on commands such as `skip`, `if-then-else` and `while-do`. First of all, all the `skip` commands are removed from the intermediate representation. Moreover, if the condition of a `if-then-else` command is always *true* then the entire statement is replaced with the first block of commands. In a similar way, if the condition is always *false* then the entire statement is replaced with the second block of commands. Furthermore, this approach is also used for `while-do` commands. That is, if the condition of a `while-do` command is always *false* then the entire statement is completely removed. However, if the condition is always *true* then an exception named *Infinite Loop* is raised. In other words, the optimizer is capable of detecting trivial infinite loops and prevents the interpretation of such programs.

## The Interpreter

Finally, the obtained intermediate representation is interpreted by the interpreter. The interpreter scan the intermediate representation using *depth-first traversal* on the n-ary tree that compose the intermediate representation itself. Before introducing the execution of the interpreter on an intermediate representation, some useful data types are defined in order to handle errors.

```
1 data Exception
2   = EmptyProgram
3   | InfiniteLoop
4   | DivisionByZero
5   | UndeclaredVariable String
6   deriving (Show)
7
8 data Result a = Ok a | Error Exception
9
10 instance Functor Result where
11   fmap f (Ok v) = Ok (f v)
12   fmap _ (Error e) = Error e
13
14 instance Applicative Result where
15   pure v = Ok v
16   (<*>) (Ok f) (Ok v) = Ok (f v)
17   (<*>) (Error e) _ = Error e
18   (<*>) _ (Error e) = Error e
19
20 instance Monad Result where
21   (>>=) (Ok v) f = f v
22   (>>=) (Error e) _ = Error e
23
24 exception :: Exception -> a
25 exception e = errorWithoutStackTrace $ show e
```

As one can see, the `Result` data type is a polymorphic type that can be either a `Ok` or a `Error`. The `Error` type also encapsulate an exception, one of the listed above. Moreover functor, applicative and monad interfaces are implemented in order to apply the needed operators directly on results. The main function that executes a block (i.e. a list of command) is reported in the following code snippet.

```
1 execute :: State -> Block -> State
2 execute s [] = s
3 execute s (Skip : cs) = execute s cs
4 execute s ((Assignment d a) : cs) =
5   case evalArithmetic s a of
6     Ok v -> execute s' cs
7     where
8       s' = insert d v s
9     Error e -> exception e
10 execute s ((Branch b cs' cs'') : cs) =
11   case evalBoolean s b of
12     Ok True -> execute s (cs' ++ cs)
13     Ok False -> execute s (cs'' ++ cs)
14     Error e -> exception e
15 execute s (c@(Loop b cs') : cs) =
16   execute s (Branch b (cs' ++ [c]) [Skip] : cs)
```



The implementation of the execution of a `while-do` resembles how it's evaluated in operational semantics, i.e. by “wrapping” it in a `if-then-else` command statement.

The implementation of the evaluation of arithmetic expressions make *heavily* use of the methods exposed by the applicative interface. Moreover, the `liftA2` function is used in order to make the functions more compact.

```

1 evalArithmetic :: State -> ArithmeticExpr -> Result Int
2 evalArithmetic _ (Constant v) = Ok v
3 evalArithmetic s (Identifier d) =
4   case search d s of
5     Just v -> Ok v
6     Nothing -> Error (UndeclaredVariable d)
7 evalArithmetic s (Add a1 a2) = liftA2 (+) v1 v2
8   where
9     v1 = evalArithmetic s a1
10    v2 = evalArithmetic s a2
11 evalArithmetic s (Sub a1 a2) = liftA2 (-) v1 v2
12   where
13    v1 = evalArithmetic s a1
14    v2 = evalArithmetic s a2
15 evalArithmetic s (Mul a1 a2) = liftA2 (*) v1 v2
16   where
17    v1 = evalArithmetic s a1
18    v2 = evalArithmetic s a2
19 evalArithmetic s (Div a1 a2) = seqM2 safeDiv v1 v2
20   where
21    v1 = evalArithmetic s a1
22    v2 = evalArithmetic s a2
23 evalArithmetic s (Mod a1 a2) = seqM2 safeMod v1 v2
24   where
25    v1 = evalArithmetic s a1
26    v2 = evalArithmetic s a2
27 evalArithmetic s (Neg a) = negate <$> v
28   where
29    v = evalArithmetic s a

```

In this implementation, I also used the functions `safeDiv`, `safeMod` and `seqM2` which implementation is showed in the following code snippet. In other words, the `seqM2` is a custom function that implements binary sequencing, similarly to the `>=` operator.

```

1 safeDiv :: Int -> Int -> Result Int
2 safeDiv _ 0 = Error DivisionByZero
3 safeDiv u v = Ok (div u v)
4
5 safeMod :: Int -> Int -> Result Int
6 safeMod _ 0 = Error DivisionByZero
7 safeMod u v = Ok (mod u v)
8
9 seqM2 :: (Monad m) => (a -> b -> m c) -> m a -> m b -> m c
10 seqM2 f x y = join $ liftA2 f x y
11
12 join :: (Monad m) => m (m a) -> m a
13 join m = m >=> id

```

The implementation of the evaluation of boolean expressions is very similar to the one for arithmetic expressions, so it's omitted.

## Conclusion

The monadic implementation of the parser give us a very simple way of concatenate and combine multiple parsers of sub-grammars. Also, it's very easy to extend the grammar of the language in order to include other commands or statements. Moreover, the use of an intermediate representation permits us to apply post-processing or optimizations and make the interpretation itself straightforward.

Future works may include the introduction of other types for variables, such as boolean types. Another extension of this work consists of adding useful information about the parsing errors, i.e. missing tokens and relevant row and column locations in the source code where the error occurred. However, other improvements can be done in the optimization step. That is, one can implement more “aggressive” optimizations of the intermediate representation based on the expected state of memory of the program during its execution.